

```

1 ; Harsh Savla & TJ Wiegman
2 ; ME 58600
3 ; 2022-09-26
4 ; initint.s
5
6 ; a subroutine to enable the external IRQ interrupt pin
7 RCC_APB2ENR      EQU 0x40021018
8 GPIOA_CRL        EQU 0x40010800
9 EXTI_IMR         EQU 0x40010400
10 EXTI_FTSR        EQU 0x4001040C
11 NVIC_ISER0       EQU 0xE000E100
12
13 ; program code
14     AREA ARMex, CODE, READONLY
15     ENTRY
16 initint PROC
17     EXPORT initint
18     ; push LR to stack
19     push {LR}
20
21     ; Enable GPIO A
22     ldr R3, =RCC_APB2ENR
23     ldrb R1, [R3]
24     orr R1, #0x04 ; enable bit 2 (0100)
25     strb R1, [R3]
26
27     ; Configure PA01
28     ldr R3, =GPIOA_CRL
29     ldrb R1, [R3]
30     and R1, #0x0F ; set PA01 to 0000
31     orr R1, #0x40 ; set PA01 to 0100
32     strb R1, [R3]
33
34     ; Unmask External Interrupt 1
35     ldr R3, =EXTI_IMR
36     mov R1, #0x02
37     str R1, [R3]
38
39     ; Set trigger to falling edge
40     ldr R3, =EXTI_FTSR
41     mov R1, #0x02
42     str R1, [R3]
43
44     ; Enable in NVIC (?)
45     ldr R3, =NVIC_ISER0
46     mov R1, #0x80
47     str R1, [R3]
48
49     ; End subroutine and go back to caller
50     pop {LR}
51     bx LR
52     ENDP
53     END

```

```

1 ; Harsh Savla & TJ Wiegman
2 ; ME 58600
3 ; 2022-09-26
4 ; inittime.s
5
6 ; a subroutine to set up the SysTick timer interrupt for every 100ms
7 STK_CTRL    EQU 0xE000E010
8 STK_LOAD    EQU 0xE000E014
9
10 ; program code
11     AREA ARMex, CODE, READONLY
12     ENTRY
13 inittime PROC
14     EXPORT inittime
15     ; push LR to stack
16     push {LR}
17
18     ; Set timer to 100 ms
19     ldr R3, =STK_LOAD
20     ldr R1, =0x249EFF
21     str R1, [R3]
22
23     ; Enable timer, systick interrupt, and 24MHz counter
24     ldr R3, =STK_CTRL
25     mov R1, #0x07
26     str R1, [R3]
27
28     ; End subroutine and go back to caller
29     pop {LR}
30     bx LR
31     ENDP
32     END

```

```

1 ; Harsh Savla & TJ Wiegman
2 ; ME 58600
3 ; 2022-09-19
4 ; serialIO.s
5
6 RCC_APB2ENR EQU 0x40021018 ; enable APB2 clock for USART1
7 GPIOA_CRH EQU 0x40010804 ; configure PA09 and PA10 for tx/rx
8 USART1_BRR EQU 0x40013808 ; configure USART1 baud rate
9 USART1_CR1 EQU 0x4001380C ; enable USART1, set parity, mode
10 USART1_SR EQU 0x40013800 ; USART1 status register
11 USART1_DR EQU 0x40013804 ; USART1 data register
12
13 USART1baud EQU 0x00D0 ; hex fraction for setting baud rate
14
15 ; allocate some RAM for bindec
16 AREA MyData, DATA, READWRITE
17 num3 SPACE 2
18 array3 SPACE 6
19
20 ; program code
21 AREA ARMex, CODE, READONLY
22 ENTRY
23 initcom PROC ; initializes serial channel 1 for asynchronous communications
24 EXPORT initcom
25 ; push LR to stack
26 push {LR}
27
28 ; turn on APB2 peripheral clock
29 ldr R3, =RCC_APB2ENR
30 ldr R1, [R3] ; save current APB2 state
31 orr R1, #0x4000
32 orr R1, #0x0005
33 str R1, [R3]
34
35 ; configure port A - PA09 output, PA10 input
36 ldr R3, =GPIOA_CRH
37 ldr R1, =0x444444B4
38 str R1, [R3]
39
40 ; configure USART1 baud rate
41 ldr R3, =USART1_BRR
42 mov R1, #USART1baud ; as close as possible to 115,200 -- impossible to get exact
   at 24MHz
43 str R1, [R3]
44
45 ; enable USART1 for 8 data bits. disable parity and interrupts
46 ldr R3, =USART1_CR1
47 mov R1, #0x200C
48 str R1, [R3]
49
50 ; End subroutine and go back to caller
51 pop {LR}
52 bx LR
53 ENDP
54

```

```

55 checkcom    PROC    ; checks to see if a character is in the data receive register. writes
0xFF to R0 if available, 0x00 otherwise
56    EXPORT checkcom
57    ; push LR to stack
58    push {LR}
59
60    ; check status register
61    ldr R3, =USART1_SR
62    ldr R1, [R3]
63    and R1, #32 ; mask out unneeded flags
64    cmp R1, #32 ; check if RXNE flag is set
65    beq ready
66
67    ; set R0 to 0x00 if not ready
68    mov R0, #0x00
69    b chEnd
70
71    ; set R0 to 0xFF if ready
72 ready mov R0, #0xFF
73
74    ; End subroutine and go back to caller
75 chEnd pop {LR}
76    bx LR
77    ENDP
78
79 getchar     PROC    ; fetches character from serial channel 1 and writes it as ASCII to R0
80    EXPORT getchar
81    ; push LR to stack
82    push {LR}
83
84    ldr R3, =USART1_DR
85    ldrb R0, [R3]
86
87    ; End subroutine and go back to caller
88    pop {LR}
89    bx LR
90    ENDP
91
92 showchar    PROC    ; checks that TXE is set, then outputs ASCII character in R0 to serial
channel 1
93    EXPORT showchar
94    ; push LR to stack
95    push {LR}
96
97    ; check status register
98 shWait ldr R3, =USART1_SR
99    ldr R1, [R3]
100    and R1, #128 ; mask out unneeded flags
101    cmp R1, #128 ; check if TXE flag is set
102    beq write
103
104    ; if DR is not ready yet
105    b shWait
106
107    ; if DR is ready

```

```

108 write    ldr R3, =USART1_DR
109          strb R0, [R3]
110
111          ; End subroutine and go back to caller
112          pop {LR}
113          bx LR
114          ENDP
115
116 bindec     PROC ; converts a 16-bit signed binary number into five decimal characters
117            (digits)
118            ; preceded by either a space or a minus sign depending on whether the signed number is
119            positive or negative
120            EXPORT bindec
121            ; push LR to stack
122            push {LR}
123
124            ; fill array with spaces
125            mov R1, #1
126            ldr R3, =array3
127            mov R0, #0x20 ; ascii character for " "
128            clrloop strb R0, [R3], #1
129            add R1, #1
130            cmp R1, #7
131            bne clrloop
132
133            ; get input number from RAM, put into R0
134            ldr R3, =num3
135            ldrh R0, [R3]
136
137            mov R1, R0
138            lsr R1, #15 ; shift first digit down to LSB
139            cmp R1, #1 ; is negative?
140            beq neg1
141
142            ; is positive
143            mov R1, #0x20 ; ascii character for " "
144            ldr R3, =array3
145            strb R1, [R3], #5
146            b binloop
147
148            ; is negative
149            neg1 mov R1, #0x2D ; ascii character for "-"
150            ldr R3, =array3
151            strb R1, [R3], #5
152            sxth R0
153            sub R0, #1
154            eor R0, #0xFFFFFFFF
155
156            ; divide by 10
157            binloop mov R4, #10
158                    udiv R1, R0, R4 ; R1 holds quotient
159                    mul R2, R1, R4
160                    sub R2, R0, R2 ; R2 holds remainder
161
162            ; convert to ASCII and store

```

```

161         add R2, #0x30
162         strb R2, [R3], #-1
163         mov R0, R1
164         cmp R0, #0
165         bne binloop
166
167         ; End subroutine and go back to caller
168         pop {LR}
169         bx LR
170         ENDP
171
172 shownum    PROC ; takes binary half-word from R0 and outputs decimal over serial
173         EXPORT shownum
174         ; push LR to stack
175         push {LR}
176
177         ; Load R0 half-word to RAM for bindec to use it
178         ldr R3, =num3
179         strh R0, [R3]
180         bl bindec
181
182         ; Loop through decimal characters until all printed
183         mov R2, #0
184 sloop     ldr R3, =array3
185         add R3, R2
186         ldrb R0, [R3]
187         bl showchar
188         add R2, #1
189         cmp R2, #6
190         bne sloop
191
192         ; Write newline afterwards, forces buffer empty
193         mov R0, #0x0A
194         bl showchar
195         mov R0, #0x0D
196         bl showchar
197
198         ; End subroutine and go back to caller
199         pop {LR}
200         bx LR
201         ENDP
202     END
203

```

```

1 ; Harsh Savla & TJ Wiegman
2 ; ME 58600
3 ; 2022-09-26
4 ; lab4main.s
5
6 EXTI_PR          EQU 0x40010414
7
8 ; allocate some RAM for clicks and outclicks
9     AREA MyData, DATA, READWRITE
10 clicks          SPACE 2
11 outclicks       SPACE 2
12
13 ; program code
14     AREA ARMex, CODE, READONLY
15     ENTRY
16 __main PROC
17     EXPORT __main
18     IMPORT initcom
19     IMPORT initint
20     IMPORT inittime
21     IMPORT shownum
22     IMPORT checkcom
23     IMPORT getchar
24
25     ; Initialize serial communications
26     bl initcom
27
28     ; Initialize clicks and outclicks to zero
29     mov R0, #0
30     ldr R3, =clicks
31     strh R0, [R3]
32     ldr R3, =outclicks
33     strh R0, [R3]
34
35     ; Set up external clock interrupt
36     bl initint
37
38     ; Set up 100ms timer
39     bl inittime
40
41     ; Main loop
42 chloop ldr R3, =outclicks
43     ldrh R1, [R3]
44     cmp R1, #0x0000
45     bne shclick
46
47     ; Check if received serial comm
48     bl checkcom
49     cmp R0, #0xFF
50     bne chloop
51
52     ; Check if received character is ESC
53     bl getchar
54     cmp R0, #0x1B ; ASCII value for ESCAPE


```

```


55         beq done
56         b chloop
57
58 shclick mov R0, R1 ; copy outclicks from R1 to R0
59         mov R1, #0
60         strh R1, [R3] ; reset outclicks to zero
61         bl shownum ; R0 still holds old outclicks value
62         b chloop
63
64 done     b done
65         ENDP
66
67 EXTI1_IRQHandler PROC
68         EXPORT EXTI1_IRQHandler
69         ; push LR to stack
70         push {LR}
71
72         ; Clear interrupt flag
73         ldr R3, =EXTI_PR
74         mov R1, #0x02
75         str R1, [R3]
76
77         ; Increase clicks value
78         ldr R3, =clicks
79         ldrh R1, [R3]
80         add R1, #1
81         strh R1, [R3]
82
83         ; End subroutine and go back to caller
84         pop {LR}
85         bx LR
86         ENDP
87
88 SysTick_Handler PROC
89         EXPORT SysTick_Handler
90         ; push LR to stack
91         push {LR}
92
93         ; Get clicks, then reset
94         ldr R3, =clicks
95         ldrh R1, [R3]
96         mov R2, #0
97         strh R2, [R3]
98
99         ; Move clicks to outclicks
100        ldr R3, =outclicks
101        strh R1, [R3]
102
103        ; End subroutine and go back to caller
104        pop {LR}
105        bx LR
106        ENDP
107    END

```



SysTick Handler



EXTI1 IRQHandler



lab4main.s




Set up serial communications

Set up external clock interrupt

Set up 100ms timer

Output 1/10 of external clock frequency

initint.s



initint.s

